



Personal Computer

Scanned by

**A Guide to Writing
BASIC Speech Programs
for the IBM PCjr**

6361978

First Edition (July 1984)

The following paragraph does not apply to the United Kingdom or any country where such provisions are inconsistent with local law: International Business Machines Corporation provides this manual "as is," without warranty of any kind, either expressed or implied, including, but not limited to, the particular purpose. IBM may make improvements and/or changes in the product(s) and/or the program(s) described in this manual at any time.

This product could include technical inaccuracies or typographical errors. Changes are periodically made to the information herein; these changes will be incorporated in new editions of the publication.

It is possible that this material may contain reference to, or information about, IBM products (machines and programs), programming, or services that are not announced in your country. Such references or information must not be construed to mean that IBM intends to announce such IBM products, programming, or services in your country.

Products are not stocked at the address below. Requests for copies of this product and for technical information about the system should be made to your authorized IBM Personal Computer dealer.

The following paragraph applies only to the United States and Puerto Rico: A Reader's Comment Form is provided at the back of this publication. If the form has been removed, address comments to: IBM Corp., Personal Computer, P.O. Box 1328-C, Boca Raton, Florida 33432. IBM may use or distribute any of the information you supply in any way it believes appropriate without incurring any obligations whatever.

© Copyright International Business Machines Corporation 1984

Scanned by
MIBB

How To Use This Book

The IBM PCjr Speech Attachment option is required by applications that use speech and some sound effects. Perhaps you purchased the Speech Attachment so that you could use those applications. You can, however, write your own programs to synthesize speech using the Speech Attachment. This book teaches some simple ways to include speech and sound effects in your BASIC programs.

You don't need to be a programming expert to use this book, but there are a few things you should understand. You must understand the fundamentals of programming in BASIC, especially variables, subroutines, and FOR-NEXT loops. If you have successfully completed the exercises in *Hands-On BASIC for the IBM PCjr*, you have all the skills you need.

This book teaches in a "hands-on" manner; that is, you learn by doing. The book asks you to type in an example program and run it; then the book discusses the program and why it works. You are encouraged to experiment with the examples, trying different ways to produce the speech and sound effects you want. The simplest ways of writing speech programs are covered first, followed by some more complex methods.

At the back of the book are Appendixes that contain reference information. You will use this information to do the exercises in the book, and you can refer to it later when you are familiar with writing speech programs.

This book teaches only the simplest ways of writing speech programs. To learn about more advanced features of the Speech Attachment, or to learn how to write speech programs in assembler language, refer to the *IBM PCjr Technical Reference*.

Notes:

Scanned by
MBB

Contents

Chapter 1. Introducing The IBM PCjr Speech Attachment	1-1
Chapter 2. How to Make PCjr Speak	2-1
Setting Up Your Speech Program Tools	2-1
Vocabulary Mode	2-3
Foreground VS. Background	2-4
Review of Vocabulary Mode Variables	2-7
If You Run Out of Memory	2-7
Record/Playback Mode	2-8
Review of Record/Playback Mode Variable	2-10
Some Notes About Compiled Basic	2-12
More About Checking Status	2-12
Resetting The Speech Attachment	2-13
Appendix A. Speech Attachment Vocabulary List ..	A-1
Appendix B. Summary of Speech Attachment Variables	B-1
Variables Used For Vocabulary Mode (P1=2 or P1=3)	B-1
Variables Used For Record/Playback Mode (P1=1)	B-2
The Status Variable (P7)	B-3

Notes:

Scanned by
MBB

Chapter 1. Introducing The IBM PCjr Speech Attachment

The Speech Attachment makes your PCjr capable of synthesizing speech and other sound effects. With the Speech Attachment, you can instruct your computer to speak words from a predefined list, or you can record and play back your own words and sounds.

The Speech Attachment comes with a vocabulary of 196 words and sounds stored on it. Those words are listed in Appendix A. When you write a program that uses words from the vocabulary list, you are using the *vocabulary mode*.

If you have a microphone attached to your Speech Attachment, you can record your own words and sounds, save them, then have your program recall them later. When you use the Speech Attachment in this manner, you are using the *record/playback mode*.

Before you do the exercises in the next chapter, be sure that the Speech Attachment is properly installed in your IBM PCjr. If you plan to use the record/playback mode, you must have a microphone installed at the rear of the Speech Attachment. For information about installing the Speech Attachment and the microphone, refer to the installation instructions that came with the Speech Attachment.

Notes:



Scanned by
MBB

Chapter 2. How to Make PCjr Speak

In this chapter we will show you how to write BASIC programs to make the PCjr synthesize speech and other sound effects. You can use either cassette BASIC or cartridge BASIC for your programs.

The Speech Attachment contains a program, called *Speech BIOS* (Basic Input/Output System), that makes the PCjr synthesize speech. In your BASIC program, you type a series of statements that tie into the Speech BIOS program and tell it how you want the computer to speak.

Setting Up Your Speech Program Tools

Note: In this section, you will be asked to save some subroutines on diskette or cassette. If you plan to use a diskette, remember to start DOS and cartridge BASIC, and be sure you have a formatted diskette before you start to type in the subroutines. If you have forgotten how to start cartridge BASIC or format a diskette, review Part One, "Taking Control," in *Hands-On BASIC for the IBM PCjr*.

Before you write a speech program, you must first type in two subroutines. The first subroutine sets up the link between your BASIC program and the Speech BIOS program. The second subroutine tells the Speech BIOS program what you want it to do, then passes control to it. Because these two subroutines are used in every speech program, and because they must always be typed in *exactly* as shown, you should save them on a diskette or cassette as your speech "toolkit". Don't be concerned about how these subroutines work: just type them as follows:

```

10000 'SUBROUTINE: SET UP LINK
10010 'SET UP THE LINK
10020 'BETWEEN SPEECH BIOS AND BASIC
10030 '
10040 DIM T%(25)
10050 DIM D%(25000)
10060 DEFINT P,T
10070 P1=0:P2=0:P3=0:P4=0:P5=0:P6=0:P7=0
10080 TMP=0:TP!=0:TALK=0:I=0
10090 TP!=VARPTR(T%(0))
10100 FOR I=0 TO 49
10110 READ TMP
10120 POKE I+TP!,TMP
10130 NEXT
10140 DEF SEG=&H50
10150 P4=PEEK(16)+PEEK(17)*256
10160 DEF SEG
10170 RETURN
10180 DATA 85,139,236,30,139,118,18,139
10190 DATA 28,139,118,16,139,4,138,227
10200 DATA 139,118,14,139,28,139,118,10
10210 DATA 139,52,86,139,118,8,139,12
10220 DATA 139,118,12,142,28,94,205,77
10230 DATA 31,139,118,6,136,4,93,202
10240 DATA 14,0

20000 'SUBROUTINE: PASS CONTROL
20010 'PASS INFORMATION AND
20020 'CONTROL TO SPEECH BIOS
20030 '
20040 TP!=VARPTR(T%(0))
20050 IF TP!>=32768! THEN TALK=TP!-65536!
20060 IF TP!<32768! THEN TALK=TP!
20070 CALL TALK(P1,P2,P3,P4,P5,P6,P7)
20080 RETURN

```

Remember to save these subroutines on a diskette or cassette.

Now that your speech toolkit is set up, you're ready to write your first speech program.

Vocabulary Mode

Let's begin with an example program that tells the PCjr to say "THE TIME IS ELEVEN P.M." If you like, you can try it now. *First, make sure that your speech toolkit subroutines are in memory.* Then, type in the following statements:

```
10 'PROGRAM TIME
20 GOSUB 10000 'SET UP LINK
30 P1=3 'VOCABULARY FOREGROUND
40 P2=1 'SELECT WORDS FROM LIST
50 P3=146 'THE
60 GOSUB 20000 'PASS CONTROL
70 P3=141 'TIME
80 GOSUB 20000 'PASS CONTROL
90 P3=96 'IS
100 GOSUB 20000 'PASS CONTROL
110 P3=76 'ELEVEN
120 GOSUB 20000 'PASS CONTROL
130 P3=111 'P.M.
140 GOSUB 20000 'PASS CONTROL
150 END
```

Run the program. The computer should speak the words "THE TIME IS ELEVEN P.M." If the program doesn't work, check your main routine and the two subroutines for typing errors. Remember, they must be typed *exactly* as shown.

Notice the variables P1, P2, and P3 in the main routine, and again in line 20070 in the subroutine. By assigning values to those variables (as in lines 30, 40, and 50, for example), then calling the subroutine (as in line 60), you tell the computer how you want it to speak.

What you have just seen is an example of playing back speech in the vocabulary mode. Remember, vocabulary mode calls upon a list of 196 words and sounds that are stored as part of the Speech BIOS program. We specified vocabulary mode in line 30 of the main routine when we assigned the value 3 to the variable

P1. By assigning the value 1 to the variable P2, we told the Speech Attachment to use a word from its vocabulary list.

The values assigned to P3 (146 in line 50 and 141 in line 70, for example) indicate the words to be spoken. If you look in the vocabulary list in Appendix A, you'll see that word number 146 is "THE" (or "THEE") and word number 141 is "TIME". You can experiment with different words by changing the value of P3. For example, if you change line 110 so that $P3=20$, then change line 130 so that $P3=56$, the computer will speak the words "THE TIME IS SEVEN A.M."

Foreground VS. Background

In our example program, we have been using the vocabulary *foreground* mode ($P1=3$). This means that the computer cannot process any other statements while it is speaking a word. The Speech Attachment can also operate in vocabulary *background* mode ($P1=2$), meaning that the computer can perform some other operations while it is speaking words.

The following example program uses vocabulary background mode. The program tells the computer to speak the numbers from "ONE" to "TEN" while it displays the word "HELLO" on the screen. In this example, lines 100 through 120 will probably look strange to you. Don't worry about them now; just type what you see, and we'll discuss them later.

Make sure that the speech toolkit subroutines are in memory. Then type:

```
10 'PROGRAM COUNT
20 GOSUB 10000 'SET UP LINK
30 P1=2 'VOCABULARY BACKGROUND
40 P3=13 'THIS VALUE WILL CHANGE
50 FOR I = 1 TO 10
60 P2=1 'USE WORD FROM LIST
70 P3=P3+1 'NEXT WORD
80 GOSUB 20000 'PASS CONTROL
90 PRINT "HELLO"
100 P2=0 'CHECK STATUS
110 GOSUB 20000 'PASS CONTROL
120 IF P7<>0 THEN 90 'IF BUSY, LOOP
130 PRINT:PRINT
140 NEXT I
150 END
```

Run the program. The computer should speak the numbers from “ONE” to “TEN” while it displays the word “HELLO” on the screen.

Let’s take a look at how the program works:

- Line 20 sets up the link to Speech BIOS.
- Line 30 sets vocabulary background mode by assigning the value 2 to P1.
- Line 40 sets P3 to 13, then line 70 adds 1 to P3, changing the value to 14. Each time the program goes through the FOR-NEXT loop (lines 50–140), line 70 increases the value of P3 by one. (Notice in the vocabulary list that the P3 values for the words "ONE" through “TEN” are 14 through 23.)
- Line 60 assigns the value 1 to P2, telling Speech BIOS to use a word from its vocabulary list.
- Line 80 passes control to Speech BIOS, telling it to follow the instructions given in P1, P2, and P3.
- While the computer is speaking, line 90 tells it to display the word “HELLO” on the screen. If you have a printer, you can change the PRINT statement to LPRINT, so the word "HELLO" is displayed on the printer instead of on the screen.

- Line 140 adds 1 to the variable I and returns to the beginning of the FOR-NEXT loop (line 50).

Lines 100–120 might seem strange, because they tell Speech BIOS to do something other than speak a word. The Speech Attachment can process only one word at a time; if it tries to process a word before the previous one is finished, an error occurs. Lines 100–120 check the status of the Speech Attachment to be sure that it has finished with one word before it attempts the next. Here's how it works:

- Line 100 assigns the value 0 to P2, which tells Speech BIOS to check its status rather than speak a word from the vocabulary list.
- Line 110 passes control to Speech BIOS.
- This time, instead of speaking a word, the Speech Attachment checks its status and assigns a value to P7.
- Line 120 tells the program to loop back to line 90 (print the word "HELLO") if the value of P7 is anything other than 0. (P7=0 means satisfactory completion.)

The program continues to run these statements until the value of P7 is 0, meaning that the Speech Attachment has completed a word and is ready to process the next word.

We have said that the Speech Attachment can process only one word at a time. There are other restrictions on what your program can do while it is running in vocabulary background mode. While the Speech Attachment is processing a word, your program cannot attempt any diskette or communications operations, nor can it issue the SYSTEM command. In any of these cases, you should check the status of the Speech Attachment to be sure that it has completed processing the word before you try another operation.

Review of Vocabulary Mode Variables

As we have seen, the values you assign to variables determine the way the Speech BIOS program works. The following is a summary of the variables used for vocabulary mode:

Variable and value	Meaning
P1=2	Sets vocabulary background mode.
P1=3	Sets vocabulary foreground mode.
P2=0	Tells the Speech Attachment to check its status.
P2=1	Tells the Speech Attachment to speak a word from its vocabulary list.
P3= <i>n</i>	Tells the Speech Attachment to speak word <i>n</i> from its vocabulary list.
P7	The Speech Attachment assigns a value to this variable, indicating the status of the Speech Attachment (P7=0 means satisfactory completion).

If You Run Out of Memory

If you write a fairly large program that uses vocabulary mode, and then try to run it, you could get an "OUT OF MEMORY" message. If that happens, simply delete line 10050 from the first subroutine in your speech toolkit. Line 10050 is needed for record/playback mode, but not for vocabulary mode.

If you delete line 10050 from the toolkit, remember to replace it before you write a program using record/playback mode.

Record/Playback Mode

If you have a microphone attached to your Speech Attachment, you can record your own words and sounds, then have your program play them back. The following program is an example of the record/playback mode.

Note: Before you try to run a program that uses record/playback mode, be sure that your microphone is properly installed at the rear of the Speech Attachment. For information about installing the microphone, refer to the installation instructions that came with the Speech Attachment.

Make sure that your speech toolkit subroutines are in memory, then type:

```
10 'PROGRAM RECORD/PLAYBACK
20 K$="" 'ASSIGN EMPTY STRING TO K$
30 GOSUB 10000 'SET UP LINK
40 P1=1 'SET RECORD/PLAYBACK MODE
50 P2=0 'RECORD FROM MICROPHONE
60 P3=5 'SET RECORD/PLAYBACK SPEED
70 P6=24000 'NUMBER OF BYTES
80 PRINT "Press any key to record."
90 K$=INKEY$:IF K$="" GO TO 90 'WAIT
100 PRINT "Recording....."
110 P5=VARPTR(D%(0)) 'MEMORY ADDRESS
120 GOSUB 20000 'PASS CONTROL
130 P2=1 'PLAY BACK SOUNDS
140 PRINT "To play back, press any key."
150 K$=INKEY$:IF K$="" GO TO 150 'WAIT
160 P5=VARPTR(D%(0)) 'MEMORY ADDRESS
170 GOSUB 20000 'PASS CONTROL
180 END
```


Run the program. After the “Press any key to record” prompt appears on the screen, you can start recording. Press any key, then speak into the microphone. You can record for 5 seconds. To play back the sounds you have just recorded, press any key again.

Note: If the program doesn’t work properly, check your program to be sure you typed it exactly as shown. Also, check to be sure that the microphone is properly installed at the rear of the Speech Attachment.

Here’s how the program works:

- Line 20 assigns the empty string to the variable K\$.
- Line 30 sets up the link to Speech BIOS.
- Line 40 sets record/playback mode by assigning the value 1 to P1.
- Line 50 assigns the value 0 to P2, indicating that you want to record words or sounds.
- Line 60 assigns the value 5 to P3, telling the Speech Attachment to record at a speed of 4800 bytes per second. Later in this section, there is a chart that shows the different values for P3 and the speeds those values represent.
- Line 70 assigns the value 24000 to P6, telling the Speech Attachment that you want to record 24000 bytes. (24000 bytes at 4800 bytes per second equals 5 seconds of recording time.)
- Line 80 prints a message on the screen, prompting you to press any key when you’re ready to begin recording, then line 90 waits for you to press a key.
- Line 100 prints a message indicating that the recording process is starting.

- Line 110 assigns to P5 the address in memory where the program will store the recorded sounds, then line 120 passes information and control to Speech BIOS.
- Line 130 assigns the value 1 to P2, indicating that you want to play back the sounds you have recorded.
- Line 140 prints a message on the screen, prompting you to press any key when you're ready to play back the recorded sounds, then line 150 waits for you to press a key.
- Line 160 assigns to P5 the address in memory where the program stored the sounds you recorded, then line 170 passes information and control to Speech BIOS.

Note: Don't be concerned about how lines 110 and 160 work; just type them as shown.

Review of Record/Playback Mode Variable

The values you assign to the Speech Attachment variables (P1, P2, and so on) have different meanings, depending on whether you're using vocabulary mode or record/playback mode. The following is a summary of the variables used for record/playback mode:

Variable and value	Meaning
P1=1	Sets record/playback mode.
P2=0	Tells the Speech Attachment to record sounds from the microphone.
P2=1	Tells the Speech Attachment to play back the sounds you've recorded.

P3

Tells the Speech Attachment the speed at which you want to record or play back words and sounds:

P3=0	1800 bytes per second
P3=1	2400 bytes per second
P3=2	3000 bytes per second
P3=3	3600 bytes per second
P3=4	4200 bytes per second
P3=5	4800 bytes per second

P6= n

Tells the Speech Attachment to record or play back n bytes.

Note: Record/playback mode uses two other variables, P4 and P5, which tell the Speech Attachment where in memory to store recorded sounds and where to get the sounds to play back. Those variables are not discussed in this book; for information about them, refer to the *IBM PCjr Technical Reference*.

To determine the length of time that you can record or play back sounds, divide the number of bytes (the value of P6) by the number of bytes per second (indicated by P3). For example, if P3=2 (3000 bytes per second) and P6=24000, your program will record and play back sound for 8 seconds.

The speech toolkit is set up so that you can record no more than 50000 bytes of sound. If you try to record more than that, your program will not work properly. The 50000-byte limit is set in line 10050 of the first toolkit subroutine:

```
10050 DIM D%(25000)
```

The number (25000) specified in that statement, multiplied by 2, equals the maximum number of bytes you can record (50000). Line 10050 reserves 50000 bytes in memory to be used only for recording sound. If you have a fairly large program, you could run out of

memory; if that happens, you'll get an "OUT OF MEMORY" message. The problem is easy to correct. Just change line 10050 to reserve less memory (that is, change 25000 to a lower number). You'll need to experiment to find the number that works best for your program.

Some Notes About Compiled Basic

If you plan to compile your program, you must make a few changes to your main routine and the toolkit subroutines. Move lines 10050 and 10060 from the subroutine to the beginning of your main routine. Then change line 20070 in the subroutine to read:

```
20070 CALL ABSOLUTE (P1,P2,P3,P4,P5,P6,P7,TALK)
```

More About Checking Status

In the example of vocabulary background mode, you saw how to check the status of the Speech Attachment by assigning the value 0 to P2, then passing control to Speech BIOS. In that example, we requested the Speech Attachment to return its status repeatedly until the value of P7 was 0.

Every time the Speech Attachment returns control to your program, it assigns a value to P7, indicating its status. (You don't have to specifically request it.) If you are having trouble running your speech programs, you can use the status information in P7 to help you find the problem. The values that the Speech Attachment can assign to P7 are:

- 0—means the the Speech Attachment has successfully completed the action you requested.

- 1—means that the Speech Attachment does not recognize the action you requested when you assigned a value to P1; that is, you assigned a value other than 0, 1, 2, or 3.
- 2—means that the Speech Attachment is busy processing a word in vocabulary mode.
- 3 or 6—means that Speech BIOS has detected a hardware problem. The next section of this book, “Resetting the Speech Attachment,” will show you how to deal with this problem.
- 4—means that the word you requested from the vocabulary list does not exist; that is, you assigned P3 a value less than 1 or greater than 196 (vocabulary mode only).
- 5—means that the Speech Attachment does not recognize the record/playback speed you requested; that is, you assigned P3 a value less than 0 or greater than 5 (record/playback mode only).

Resetting The Speech Attachment

If the Speech BIOS program detects a hardware problem, it assigns a value of either 3 or 6 to P7. If that happens, you must reset the attachment before you attempt another speech operation. The following example shows how you could add a few statements (lines 121—126) to reset the Speech Attachment in our record/playback example program:

```

. (Lines 10--100 as shown in example program)
.
110 P5=VARPTR(D%(0)) 'MEMORY ADDRESS
120 GOSUB 20000 'PASS CONTROL
121 IF P7<>3 OR P7<>6 THEN GO TO 130
122 P1=0 'RESET SPEECH ATTACHMENT
123 GOSUB 20000 'PASS CONTROL
124 IF P7<>3 OR P7<>6 THEN GO TO 110
125 PRINT "ERROR IN SPEECH ATTACHMENT"
126 STOP 'STOP THE PROGRAM
130 P2=1 'PLAY BACK SOUNDS
.
. (Lines 140--180 as shown in example program)
.

```

Here's how the example works:

- Line 121 checks the value that the Speech Attachment has assigned to P7. If it is not 3 or 6, then the program continues as normal; Speech BIOS did not detect a hardware problem.
- If the Speech Attachment assigned a value of 3 or 6 to P7, line 122 assigns the value 0 to P1, telling Speech BIOS to reset the Speech Attachment.
- Line 123 passes information and control to the Speech Attachment, and the attachment is reset.
- After it resets the attachment, Speech BIOS again returns its status in P7. If that value is not 3 or 6, line 124 returns to line 110, so the program can try again to record. If the value returned in P7 is 3 or 6, the program prints an error message (line 125), then stops (line 126).

If the problem persists, your program could get into an infinite loop in lines 110—126, with the message “Recording.....” on the screen. Use the Break function to get out of the loop.

If your program stops or gets into a loop, there is a problem with the Speech Attachment. Follow the diagnostic instructions in the *Guide to Operations* to determine what you should do next.

Scanned by
MBB

Notes:



Scanned by
MBB

Appendix A. Speech Attachment Vocabulary List

This Appendix lists the words that are stored on the Speech Attachment as part of Speech BIOS. Beside each word is the value you assign to the variable P3 to tell the Speech Attachment to speak that word.

Remember, you can use this vocabulary list only when you are using the vocabulary mode (P1=2 or P1=3).

1 danger	67 cents	131 teen
2 time has expired	68 control	132 true
3 laughing	69 date	133 to-
4 get ready	70 disk	134 -ty
5 go	71 day	135 this
6 up	72 dollar	136 twelve
7 down	73 down	137 thousand
8 left	74 do	138 that
9 right	75 excellent	139 than
10 warning	76 eleven	140 then
11 well done	77 ez	141 time
12 gotcha	78 ed (past tense morpheme)	142 type
13 zero	79 echo	143 thing
14 one	80 equals	144 try
15 two	81 enter	145 turn
16 three	82 end	146 thee
17 four	83 first	147 twenty
18 five	84 from	148 word
19 six	85 false	149 white
20 seven	86 file	150 wait
21 eight	87 fif-	151 wrong
22 nine	88 function	152 what
23 ten	89 go	153 yes
24 a	90 green	154 you
25 b	91 good	155 yellow
26 c	92 hundred	156 year
27 d	93 hold	157 your
28 e	94 hour	158 space
29 f	95 home	159 delete
30 g	96 is	160 page
31 h	97 it	161 cursor
32 i	98 key	162 name
33 j		163 letter

34 k	99 last	164 board
35 l	100 lose	165 any
36 m	101 list	166 sign
37 n	102 less	167 spell
38 o	103 left	168 win
39 p	104 ok	169 pause
40 q	105 or	170 bar
41 r	106 period	171 insert
42 s	107 plus	172 look
43 t	108 please	173 lock
44 u	109 program	174 3 frames of silence
45 v	110 press	175 minus
46 w	111 p.m.	176 million
47 x	112 per	177 month
48 y	113 point	178 minute
49 z	114 run	179 move
50 an	115 read	180 no
51 again	116 red	181 negative
52 alt	117 right	182 number
53 add	118 release	183 not
54 am	119 start	184 alternate
55 are	120 stop	185 up
56 a.m.	121 s (plural morpheme)	186 -ing
57 ahead		187 chime 1
58 answer	122 save	188 bat hitting ball
59 back	123 second	189 ball being caught
60 by	124 sorry	190 gunshot
61 brake	125 screen	191 laser
62 at	126 score	192 phaser
63 as	127 select	193 ball hitting wall
64 and	128 th	194 tic 2
65 code	129 third	195 toc 2
66 computer	130 thir-	196 fast chime

Appendix B. Summary of Speech Attachment Variables

This Appendix lists the variables you use when you write speech programs. It summarizes the values you can assign to those variables, and what each value means.

Variable	Value	Meaning
P1	0	Reset the Speech Attachment
	1	Set record/playback mode
	2	Set vocabulary background mode
	3	Set vocabulary foreground mode

Variables Used For Vocabulary Mode (P1=2 or P1=3)

Variable	Value	Meaning
P2	0	Check the status of the Speech Attachment
	1	Speak a word from the vocabulary list
P3	n	Speak word number n from the vocabulary list

Variables Used For Record/Playback Mode (P1=1)

Variable	Value	Meaning
P2	0	Record sounds using the microphone
	1	Play back the recorded sounds
P3		Tells the Speech Attachment the speed at which to record or play back sounds:
	0	1800 bytes per second
	1	2400 bytes per second
	2	3000 bytes per second
	3	3600 bytes per second
	4	4200 bytes per second
5	4800 bytes per second	
P6	<i>n</i>	Tells the Speech Attachment to record or play back <i>n</i> bytes.

Note: Record/playback mode uses two other variables, P4 and P5, which tell the Speech Attachment where in memory to store recorded sounds and where to get the sounds to play back. Those two variables are not discussed in this book; for information about them, refer to the *IBM PCjr Technical Reference*.

The Status Variable (P7)

The variable P7 is different from the other Speech Attachment variables. You use the variables P1 through P6 to pass information to the Speech Attachment; P7 passes information back to you.

Variable	Value	Meaning
P7	0	The Speech Attachment has successfully completed the action you requested
	1	The Speech Attachment does not recognize the value you assigned to P1
	2	The Speech Attachment is busy processing a word in vocabulary mode
	3	Speech BIOS has detected a hardware problem in the Speech Attachment
	4	(Vocabulary mode only) The Speech Attachment does not recognize the value you assigned to P3
	5	(Record/playback mode only) The Speech Attachment does not recognize the value you assigned to P3
	6	Speech BIOS has detected a hardware problem in the Speech Attachment

For more detailed information about these variables, as well as information about variables not covered in this book, refer to the *IBM PCjr Technical Reference*.

Notes:



Scanned by
MBB



Reader's Comment Form

**A Guide to Writing
BASIC Speech Programs
for the IBM PCjr**

6361978

Your comments assist us in improving the usefulness of our publication; they are an important part of the input used for revisions.

IBM may use and distribute any of the information you supply in any way it believes appropriate without incurring any obligation whatever. You may, of course, continue to use the information you supply.

Please do not use this form for technical questions regarding the IBM Personal Computer or programs for the IBM Personal Computer, or for requests for additional publications; this only delays the response. Instead, direct your inquiries or request to your authorized IBM Personal Computer dealer.


Comments:

A1B3B



NO POSTAGE
NECESSARY
IF MAILED
IN THE
UNITED STATES

BUSINESS REPLY MAIL
FIRST CLASS PERMIT NO. 321 BOCA RATON, FLORIDA 33432



POSTAGE WILL BE PAID BY ADDRESSEE

IBM PERSONAL COMPUTER
SALES & SERVICE
P.O. BOX 1328-C
BOCA RATON, FLORIDA 33432



Fold here